

Akronym-Tabelle in einem WORD-Dokument aktualisieren

Dr. Volker Thormählen, Januar 2021

Inhalt

1	Aufgabe	2
2	Schreibweisen eines Akronyms	2
3	Lösungsschritte	2
4	VBA-Prozeduren	3
5	Ergebnis.....	7
6	Literaturverzeichnis.....	7

Listings

Listing 1:	Unterprozedur "Update_Acronym_Table" zur Aktualisierung der Akronym-Tabelle.....	4
Listing 2:	Funktionsprozedur "GetDefinition" zur automatischen Ermittlung der Definition eines Akronyms	5
Listing 3:	Funktionsprozedur "InputDefinition" zur manuellen Erfassung der Definition eines Akronyms	6
Listing 4:	Funktionsprozedur "GetCellContent" zur Bestimmung des Inhalts einer Tabellenzelle.....	7

1 Aufgabe

Das Aktualisieren einer bereits vorhandenen Tabelle mit Akronymen¹ in einem WORD-Dokument ist notwendig, wenn sich durch nachträgliche Überarbeitung oder Ergänzung die Seitennummern und/oder die darin evtl. enthaltenen Definitionen ändern.

Angenommen, eine 3-spaltige Tabelle mit den Akronymen eines WORD-Dokuments ist bereits vorhanden, entweder manuell erstellt oder mit einer VBA²-Prozedur generiert (vgl. [1]). Die Akronym-Tabelle ist beispielsweise wie folgt aufgebaut und gefüllt (s. Tabelle 1):

Akronym	Definition	Erstes Vorkommen auf Seite ...
AUDI		
PORSCHE		
VW		

Tabelle 1: Beispiel einer Akronym-Tabelle mit Automarken

Für das jeweils erste Vorkommen eines Akronymes (s. 1. Spalte) soll die dazugehörige Seitennummer (s. 3. Spalte) im aktuellen Dokument automatisch ermittelt werden.

Außerdem soll die Spalte „Definition“ (s. 2. Spalte) je Akronym automatisch bzw. halbautomatisch gefüllt werden.

2 Schreibweisen eines Akronymes

Folgende drei Schreibweisen eines Akronymes im Text eines WORD-Dokuments werden erwartet:

- Akronym (Definition), z. B. ... VW (Volkswagen) ...
- (Definition) Akronym, z. B. ... (Volkswagen) VW ...
- Akronym ohne Definition, z. B. ... VW ...

Es ist hilfreich, die Definition in runden Klammern anzugeben, wenn ein Akronym zum ersten Mal erwähnt wird. Bei den beiden zuerst genannten Schreibweisen kann die Definition automatisch erkannt und in die 2. Spalte der Akronym-Tabelle eingetragen werden. Sonst ist zuvor die manuelle Erfassung der Definition des betreffenden Akronymes notwendig.

Die jeweilige Definition in runden Klammern kann aus dem Dokument entfernt werden, sobald sie in die Tabelle der Akronyme eingesetzt wurde.

3 Lösungsschritte

Die Lösung erfolgt mit der VBA-Prozedur *“Update_Acronym_Table“* in vier Schritten:

Die Akronym-Tabelle ...

- ...im aktuellen Dokument finden
- ...aufsteigend nach Akronymen sortieren (1. Spalte, aber ohne Spaltenkopf)
- ...mit Definitionen automatisch bzw. halbautomatisch ergänzen (2. Spalte)
- ...mit den Seitennummern des ersten Vorkommens eines Akronymes aktualisieren (3. Spalte)

¹ Ein Akronym ist ein aus den Anfangsbuchstaben mehrerer Wörter gebildetes Kurzwort.

² Das Akronym **VBA** steht für **V**isual **B**asic for **A**pplications, einer proprietären Skriptsprache von Microsoft.

4 VBA-Prozeduren

Listing 1 enthält den VBA-Quellcode zur Lösung der beschriebenen Aufgaben. Die entsprechende Unterprozedur „**Update_Acronym_Table**“ benötigt drei Funktionsprozeduren:

- **“GetDefinition”** zur automatischen Ermittlung der Definition eines Akronyms (s. Listing 2)
- **“InputDefinition”** zur manuellen Erfassung der Definition eines Akronyms (s. Listing 3)
- **“GetCellContent”** zur Bestimmung des Inhalts einer Tabellenzelle (s. Listing 4)

Der Quellcode der vier VBA-Prozeduren wird im Folgenden mit Kommentaren vorgestellt.

```

1 Sub Update_Acronym_Table()
2   ' Aufgaben: 1. Akronym-Tabelle im aktuellen Dokument finden
3   '           2. Akronym-Tabelle aufsteigend nach Akronymen sortieren
4   '           3. Akronym-Tabelle mit Definitionen ergänzen
5   '           4. Akronym-Tabelle mit Seitennummern aktualisieren
6   ' Variablen definieren:
7   Dim objDoc      As Document      ' Zieldokument
8   Dim objTbl      As Table         ' Akronym-Tabelle im Zieldokument
9   Dim intTblNum   As Integer       ' Zählnummer der vorhandenen Akronym-Tabelle
10  Dim strColHdr   As String        ' Kopf der 1. Spalte d. Akronym-Tabelle
11  Dim strAcronym  As String        ' Akronym der 1. Spalte d. Akronym-Tabelle
12
13  On Error GoTo Error_Point      ' Fehlerhandling
14  Application.ScreenUpdating = False ' Bildschirmaktualisierung ausschalten
15  Set objDoc = ActiveDocument    ' Aktuelles Dokument als Zieldokument bestimmen
16  objDoc.Range.Select           ' Zieldokument auswählen
17  Selection.HomeKey unit:=wdStory ' Zum Anfang des Zieldokuments springen
18  ' 1. Akronym-Tabelle im aktuellen Dokument finden
19  intTblNum = 0
20  For intTblNum = 1 To objDoc.Tables.Count
21    Set objTbl = objDoc.Tables(intTblNum)
22    strColHdr = GetCellContent(objTbl.Cell(1, 1))
23    If strColHdr Like "A?ronym" Then
24      Exit For
25    End If
26  Next intTblNum
27  If intTblNum = 0 Then
28    MsgBox "Abbruch: Akronym-Tabelle im aktuellen Dokument nicht gefunden!", vbInformation
29    GoTo Exit_Point
30  End If
31  ' 2. Akronym-Tabelle aufsteigend sortieren nach Akronymen in der 1. Spalte
32  objTbl.Sort ExcludeHeader:=True, _
33             FieldNumber:="Column 1", _
34             SortFieldType:=wdSortFieldAlphanumeric, _
35             SortOrder:=wdSortOrderAscending
36
37  ' Die sortierten Akronyme in Spalte 1 der Akronym-Tabelle ab Zeile 2 durchlaufen:
38  For lngRow = 2 To objTbl.Rows.Count
39    strAcronym = GetCellContent(objTbl.Rows(lngRow).Cells(1))
40    If Len(strAcronym) > 0 Then
41      objDoc.Range.Select
42      Selection.HomeKey unit:=wdStory
43      With Selection.Find
44        .Forward = True
45        .ClearFormatting
46        .Text = strAcronym
47        .MatchWholeWord = True
48        .MatchCase = False
49        .Wrap = wdFindContinue
50        .Execute
51      End With
52      If Selection.Find.Found = True Then
53        ' 3. Akronym-Tabelle mit der Definition des jeweils gefundenen Akronyms aktualisieren
54        strDefinition = GetDefinition(strAcronym)
55        objTbl.Rows(lngRow).Cells(2).Range.Text = strDefinition
56        ' 4. Akronym-Tabelle mit Seitennummer der jeweiligen Fundstelle aktualisieren
57        objTbl.Rows(lngRow).Cells(3).Range.Text = Selection.Information(wdActiveEndPageNumber)
58      End If
59    End If
60  Next lngRow
61  MsgBox "Normales Ende der Prozedur."
62 Exit_Point:
63  Application.ScreenUpdating = True ' Bildschirmaktualisierung einschalten
64  Selection.HomeKey unit:=wdStory   ' Zum Anfang des Zieldokuments springen
65  ' Objektvariablen freigeben
66  Set objTbl = Nothing
67  Set objDoc = Nothing
68  Exit Sub
69 Error_Point:
70  MsgBox "Laufzeitfehler " & Err.Number & " " & Err.Description, vbCritical, "Update_Acronym_Table"
71  Resume Exit_Point
72 End Sub

```

Listing 1: Unterprozedur "Update_Acronym_Table" zur Aktualisierung der Akronym-Tabelle

```

1 Function GetDefinition(strAcromym As String) As String
2   ' Aufgabe: Definition eines Akronyms in runden Klammern im Text suchen,
3   ' je nach Schreibweise entweder NACH oder VOR dem dazugehörigen Akronym.
4   Const conLeft As String = "(" ' linke runde Klammer
5   Const conRight As String = ")" ' rechte runde Klammer
6   Dim lngStart As Long ' Startposition einer Definition im Text
7   Dim lngEnd As Long ' Endposition einer Definition im Text
8   ' Anfangswert setzen für die Definition eines Akronyms
9   strDefinition = vbNullString
10  ' Definition des übergebenen Akronyms im Text suchen je nach Schreibweise
11  With Selection
12    ' Schreibweise im Text: ACRONYM (Definition)
13    .MoveRight unit:=wdWord
14    .MoveRight unit:=wdCharacter, Extend:=wdExtend
15    If .Text = conLeft Then
16      lngStart = .Start
17      Do While .Text <> conRight
18        strDefinition = strDefinition & .Text
19        .Collapse Direction:=wdCollapseEnd
20        .MoveRight unit:=wdCharacter, Extend:=wdExtend
21      Loop
22    End If
23    .Collapse Direction:=wdCollapseEnd
24    lngEnd = .End
25    If Left(strDefinition, 1) = conLeft Then
26      strDefinition = Mid(strDefinition, 2, Len(strDefinition))
27      ' Definition in runden Klammern aus d. Text d. Dokuments entfernen.
28      .SetRange Start:=lngStart, End:=lngEnd + 1
29      .Delete
30    End If
31
32    ' Schreibweise im Text: (Definition) ACRONYM
33    If Len(strDefinition) = 0 Then ' Wenn Definition/Bedeutung noch nicht gefunden, dann ...
34      .MoveLeft unit:=wdCharacter, Count:=1, Extend:=wdMove
35      .MoveLeft unit:=wdWord
36      lngEnd = .End
37      .MoveLeft unit:=wdCharacter, Count:=2, Extend:=wdMove
38      If .Text = conRight Then
39        Do While .Text <> conLeft
40          strDefinition = strDefinition & .Text
41          .Collapse Direction:=wdCollapseStart
42          .MoveLeft unit:=wdCharacter, Extend:=wdExtend
43        Loop
44      End If
45      .Collapse Direction:=wdCollapseEnd
46
47      If Left(strDefinition, 1) = conRight Then
48        lngStart = .Start
49        strDefinition = StrReverse(Mid(strDefinition, 2, Len(strDefinition)))
50        ' Definition in runden Klammern aus d. Text d. Dokuments entfernen
51        .SetRange Start:=lngStart - 1, End:=lngEnd
52        .Delete
53      End If
54    End If
55  End With
56
57  If strDefinition = vbNullString Then
58    strDefinition = InputDefinition(strAcromym)
59  End If
60  GetDefinition = strDefinition
61 End Function

```

Listing 2: Funktionsprozedur "GetDefinition" zur automatischen Ermittlung der Definition eines Akronyms

```

1  Function InputDefinition(strAcronym) As Variant
2      ' Aufgabe: Definition eines Akronyms manuell erfassen.
3      Dim strPrompt, strTitle, strDef As String
4      strPrompt = "Definition eingeben für das Akronym: " & strAcronym
5      strTitle = "Definition erfassen."
6      Do
7          strDef = InputBox(strPrompt, strTitle)
8          If strDef = vbNullString Then
9              MsgBox "Abbruch durch den Benutzer!", vbInformation
10             Exit Do
11         ElseIf Len(strDef) < 3 Then
12             MsgBox "Die Definition für das Akronym " & strAcronym & " ist zu kurz, nur " & _
13                 Len(strDef) & " Zeichen eingegeben!", vbInformation
14         End If
15     Loop Until Len(strDef) > 2
16     InputDefinition = strDef
17 End Function

```

Listing 3: Funktionsprozedur "InputDefinition" zur manuellen Erfassung der Definition eines Akronyms

```

1 Function GetCellContent(ByRef objCell As Cell) As String
2   ' Aufgabe: Inhalt (d. h. Zeichenkette) der übergebenen Zelle bestimmen.
3   Dim objRng As Range
4   Set objRng = objCell.Range
5   objRng.End = objRng.End - 1
6   GetCellContent = Trim(objRng.Text)
7   Set objRng = Nothing
8 End Function

```

Listing 4: Funktionsprozedur "GetCellContent" zur Bestimmung des Inhalts einer Tabellenzelle

5 Ergebnis

- Die 2. Spalte der Akronym-Tabelle wird automatisch gefüllt mit der vorgefundenen Definition des jeweiligen Akronyms im aktuellen Dokument oder – falls nicht gefunden – manuell erfasst.
- Die 3. Spalte der Akronym-Tabelle wird *automatisch* gefüllt mit der Seitennummer des ersten Vorkommens des jeweiligen Akronyms im aktuellen Dokument.
- Die Definition von Akronymen in runden Klammern im Text des Dokuments wird entfernt, um überflüssige Informationen zu vermeiden.

6 Literaturverzeichnis

- [1] L. Fredborg, „Extrahieren von ACRONYMS in ein neues Dokument,“ 2008. [Online]. Available: <https://www.thedoctools.com/word-macros-tips/word-macros/extract-acronyms-to-new-document/>. [Zugriff am 19 01 2021].
- [2] V. Thormählen, „Abkürzungsverzeichnis für ein Word-Dokument automatisch erstellen: Akronyme tabellieren,“ 2021 02 02. [Online]. Available: https://www.dr-thormaehlen.de/WORD/Abkürzungsverzeichnis_erstellen.pdf. [Zugriff am 2021 02 02].